




Anders: Modal Homotopy Type System

Namdak Tonpa  

Groupoid Infinity

Siegmentation Fault 

Groupoid Infinity

Abstract

Here is presented a reincarnation of **cubicaltt** called **anders**.

2012 ACM Subject Classification Theory of computation → Lambda calculus

Keywords and phrases Homotopy Type System, Cubical Type Theory, Modal HoTT

1 Introduction

Anders is a Modal HoTT proof assistant based on: classical MLTT-80 [8] with 0, 1, 2, W types; CCHM [3] in CHM [4] flavour as cubical type system with hcomp/transp Kan operations; HTS [10] strict equality on pretypes; de Rham [2] stack modality primitives. We tend not to touch general recursive higher inductive schemes yet, instead we will try to express as much HIT as possible through W, Coequalizer and HubSpokes Disc in the style of HoTT/Coq homotopy library and Three-HIT theorem.

The HTS language proposed by Voevodsky exposes two different presheaf models of type theory: the inner one is homotopy type system presheaf that models HoTT and the outer one is traditional Martin-Löf type system presheaf that models set theory with UIP. The motivation behind this doubling is to have an ability to express semisimplicial types. Theoretical work on merging inner and outer languages was continued in 2LTT [1].

Installation. While we are on our road to Lean-like tactic language, currently we are at the stage of regular cubical HTS type checker with CHM-style primitives. You may try it from Github sources: `groupoid/anders`¹ or install through OPAM package manager. Main commands are **check** (to check a program) and **repl** (to enter the proof shell).

```
$ opam install anders
```

Anders is fast, idiomatic and educational (think of optimized Mini-TT). We carefully draw the favourite Lean-compatible syntax to fit 200 LOC in Menhir. The CHM kernel is 1K LOC. Whole Anders compiles under 2 seconds and checks all the base library under 1 second [17-8700]. Anders proof assistant as Homotopy Type System comes with its own Homotopy Library².

2 Syntax

The syntax resembles original syntax of the reference CCHM type checker `cubicaltt`, is slightly compatible with Lean syntax and contains the full set of Cubical Agda [9] primitives (except generic higher inductive schemes).

Here is given the mathematical pseudo-code notation of the language expressions that come immediately after parsing. The core syntax definition of HTS language corresponds to the type defined in OCaml module:

¹ <https://github.com/groupoid/anders/>

² <https://homotopy.dev/library/>

```

cosmos :=  $\mathbf{U}_j \mid \mathbf{V}_k$ 
var := var name  $\mid$  hole
pi :=  $\Pi$  name E E  $\mid \lambda$  name E E  $\mid E$  E
sigma :=  $\Sigma$  name E E  $\mid (E, E) \mid E.1 \mid E.2$ 
0 := 0  $\mid \mathbf{ind}_0$  E E E
1 := 1  $\mid \star \mid \mathbf{ind}_1$  E E E
2 := 2  $\mid 0_2 \mid 1_2 \mid \mathbf{ind}_2$  E E E
W := W ident E E  $\mid \mathbf{sup}$  E E  $\mid \mathbf{ind}_W$  E E
id := Id E  $\mid \mathbf{ref}$  E  $\mid \mathbf{id}_J$  E
path := Path E  $\mid E^i \mid E @ E$ 
I := I  $\mid 0 \mid 1 \mid E \bigvee E \mid E \bigwedge E \mid \neg E$ 
part := Partial E E  $\mid [(E = I) \rightarrow E, \dots]$ 
sub := inc E  $\mid \mathbf{ouc}$  E  $\mid E [I \mapsto E]$ 
kan := transp E E  $\mid \mathbf{hcomp}$  E
glue := Glue E  $\mid \mathbf{glue}$  E  $\mid \mathbf{unglue}$  E E
Im := Im E  $\mid \mathbf{Inf}$  E  $\mid \mathbf{Join}$  E  $\mid \mathbf{ind}_{Im}$  E E

```

Further Menhir BNF notation will be used to describe the top-level language E parser.

```

E := cosmos  $\mid$  var  $\mid$  MLTT  $\mid$  CCHM  $\mid$  Im
CCHM := path  $\mid$  I  $\mid$  part  $\mid$  sub  $\mid$  kan  $\mid$  glue
MLTT := pi  $\mid$  sigma  $\mid$  id

```

Keywords. The words of a top-level language, file or repl, consist of keywords or identifiers. The keywords are following: **module**, **where**, **import**, **option**, **def**, **axiom**, **postulate**, **theorem**, **(**, **)**, **[**, **]**, **<**, **>**, **/**, **.1**, **.2**, Π , Σ , **,**, λ , **V**, \bigvee , \bigwedge , **-**, **+**, **@**, **PathP**, **transp**, **hcomp**, **zero**, **one**, **Partial**, **inc**, **\times** , **\rightarrow** , **$:$** , **$=$** , **\mapsto** , **U**, **ouc**, **interval**, **inductive**, **Glue**, **glue**, **unglue**.

Identifiers. Identifiers support UTF-8. Identifiers couldn't start with **:**, **-**, **\rightarrow** . Sample identifiers: **\neg -of- \bigvee** , **$1 \rightarrow 1$** , **is-?**, **=**, **$\$ \sim$** !005x, **$\infty$** , **$x \rightarrow \text{Nat}$** .

Modules. Modules represent files with declarations. More accurate, BNF notation of module consists of imports, options and declarations.

```

menhir
  start <Module.file> file
  start <Module.command> repl
  repl : COLON IDENT exp1 EOF  $\mid$  COLON IDENT EOF  $\mid$  exp0 EOF  $\mid$  EOF
  file : MODULE IDENT WHERE line* EOF
  path : IDENT
  line : IMPORT path+  $\mid$  OPTION IDENT IDENT  $\mid$  declarations

```

Imports. The import construction supports file folder structure (without file extensions) by using reserved symbol **/** for hierarchy walking.

Options. Each option holds bool value. Language supports following options: 1) girard (enables **U** : **U**); 2) pre-eval (normalization cache); 3) impredicative (infinite hierarchy with impredicativity rule); In Anders you can enable or disable language core types, adjust syntaxes or tune inner variables of the type checker.

Declarations. Language supports following top level declarations: 1) axiom (non-computable declaration that breaks normalization); 2) postulate (alternative or inverted axiom that can preserve consistency); 3) definition (almost any explicit term or type in type theory); 4) lemma (helper in big game); 5) theorem (something valuable or complex enough).

```

axiom isProp (A : U) : U
def isSet (A : U) : U :=  $\Pi$  (a b : A) (x y : Path A a b), Path (Path A a b) x y

```

Sample declarations. For example, signature $\text{isProp } (A : U)$ of type U could be defined as normalization-blocking axiom without proof-term or by providing proof-term as definition.

In this example $(A : U)$, $(a b : A)$ and $(x y : \text{Path } A \ a \ b)$ are called telescopes. Each telescope consists of a series of lenses or empty. Each lense provides a set of variables of the same type. Telescope defines parameters of a declaration. Types in a telescope, type of a declaration and a proof-terms are a language expressions exp1 .

```

menhir
  ident : IRREF | IDENT
  lense : LPARENS ident+ COLON exp1 RPARENS
  telescope : lense telescope
  params : telescope | []
  declarations :
    | DEF IDENT params DEFEQ exp1
    | DEF IDENT params COLON exp1 DEFEQ exp1
    | AXIOM IDENT params COLON exp1

```

Expressions. All atomic language expressions are grouped by four categories: exp0 (pair constructions), exp1 (non neutral constructions), exp2 (path and π applications), exp3 (neutral constructions).

```

menhir
  face : LPARENS IDENT IDENT IDENT RPARENS
  part : face+ ARROW exp1
  exp0 : exp1 COMMA exp0 | exp1
  exp1 : LSQ separated(COMMA, part) RSQ
    | LAM telescope COMMA exp1 | PI telescope COMMA exp1
    | SIGMA telescope COMMA exp1 | LSQ IRREF ARROW exp1 RSQ
    | LT ident+ GT exp1 | exp2 ARROW exp1
    | exp2 PROD exp1 | exp2

```

The LR parsers demand to define exp1 as expressions that cannot be used (without a parens enclosure) as a right part of left-associative application for both Path and Pi lambdas. Universe indices U_j (inner fibrant), V_k (outer pretypes) and S (outer strict omega) are using unicode subscript letters that are already processed in lexer.

```

menhir
  exp2 : exp2 exp3 | exp2 APPFORMULA exp3 | exp3
  exp3 : LPARENS exp0 RPARENS LSQ exp0 MAP exp0 RSQ
    | HOLE | PRE | KAN | IDJ exp3
    | exp3 FST | exp3 SND | NEGATE exp3 | INC exp3
    | exp3 AND exp3 | exp3 OR exp3 | ID exp3 | REF exp3
    | OUC exp3 | PATHP exp3 | PARTIAL exp3 | IDENT
    | IDENT LSQ exp0 MAP exp0 RSQ | HCOMP exp3
    | LPARENS exp0 RPARENS | TRANSP exp3 exp3

```

3 Semantics

The idea is to have a unified layered type checker, so you can disable/enable any MLTT-style inference, assign types to universes and enable/disable hierarchies. This will be done by providing linking API for pluggable presheaf modules. We selected 5 levels of type checker awareness from universes and pure type systems up to synthetic language of homotopy type theory. Each layer corresponds to its presheaves with separate configuration for universe hierarchies. We want to mention here with homage to its authors all categorical models of

```

inductive lang : U :=
  | UNI: cosmos → lang
  | PI: pure lang → lang
  | SIGMA: total lang → lang
  | ID: uip lang → lang
  | PATH: homotopy lang → lang
  | GLUE: gluing lang → lang
  | HIT: hit lang → lang

```

dependent type theory: Comprehension Categories (Grothendieck, Jacobs), LCCC (Seely), D-Categories and CwA (Cartmell), CwF (Dybjer), C-Systems (Voevodsky), Natural Models (Awodey). While we can build some transports between them, we leave this exercise for our mathematical components library. We will use here the Coquand's notation for Presheaf Type Theories in terms of restriction maps.

3.1 Universe Hierarchies

Language supports Agda-style hierarchy of universes: prop, fibrant (U), interval pretypes (V) and strict omega with explicit level manipulation. All universes are bounded with preorder

$$Prop_i \prec Fibrant_j \prec Pretypes_k \prec Strict_l, \quad (1)$$

in which i, j, k, l are bounded with equation:

$$i < j < k < l. \quad (2)$$

Large elimination to upper universes is prohibited. This is extendable to Agda model:

```

inductive cosmos : U :=
  | prop: nat → cosmos
  | fibrant: nat
  | pretypes: nat
  | strict: nat
  | omega
  | lock

```

The **anders** model contains only fibrant U_j and pretypes V_k universe hierarchies.

3.2 Dependent Types

► **Definition 1** (Type). *A type is interpreted as a presheaf A , a family of sets A_I with restriction maps $u \mapsto u f, A_I \rightarrow A_J$ for $f : J \rightarrow I$. A dependent type B on A is interpreted by a presheaf on category of elements of A : the objects are pairs (I, u) with $u : A_I$ and morphisms $f : (J, v) \rightarrow (I, u)$ are maps $f : J \rightarrow I$ such that $v = u f$. A dependent type B is thus given by a family of sets $B(I, u)$ and restriction maps $B(I, u) \rightarrow B(J, u f)$.*

We think of A as a type and B as a family of presheaves $B(x)$ varying $x : A$. The operation $\Pi(x : A)B(x)$ generalizes the semantics of implication in a Kripke model.

► **Definition 2** (Pi). *An element $w : [\Pi(x : A)B(x)](I)$ is a family of functions $w_f : \Pi(u : A(J))B(J, u)$ for $f : J \rightarrow I$ such that $(w_f u)g = w_{fg}(u g)$ when $u : A(J)$ and $g : K \rightarrow J$.*

```
inductive pure (lang : U) : U :=
  | pi: name → nat → lang → lang → pure lang
  | lambda: name → nat → lang → lang
  | app: lang → lang
```

► **Definition 3** (Sigma). *The set $\Sigma(x : A)B(x)$ is the set of pairs (u, v) when $u : A(I), v : B(I, u)$ and restriction map $(u, v) f = (u f, v f)$.*

```
inductive total (lang : U) : U :=
  | sigma: name → lang → total lang
  | pair: lang → lang
  | fst: lang
  | snd: lang
```

The presheaf with only Pi and Sigma is called **MLTT-72** [6]. Its internalization in **anders** is as follows:

```
def MLTT-72 (A : U) (B : A → U) : U := Σ
  (Π-form1 : U)
  (Π-ctor1 : Π A B → Π A B)
  (Π-elim1 : Π A B → Π A B)
  (Π-comp1 : (a : A) (f : Π A B), Π-elim1 (Π-ctor1 f) a = f a)
  (Π-comp2 : (a : A) (f : Π A B), f = λ (x : A), f x)
  (Σ-form1 : U)
  (Σ-ctor1 : Π (a : A) (b : B a), Sigma A B)
  (Σ-elim1 : Π (p : Sigma A B), A)
  (Σ-elim2 : Π (p : Sigma A B), B (pr1 A B p))
  (Σ-comp1 : Π (a : A) (b : B a), a = Σ-elim1 (Σ-ctor1 a b))
  (Σ-comp2 : Π (a : A) (b : B a), b = Σ-elim2 (a, b))
  (Σ-comp3 : Π (p : Sigma A B), p = (pr1 A B p, pr2 A B p)), 1
```

3.3 Path Equality

The fundamental development of equality inside MLTT provers led us to the notion of ∞ -groupoid as spaces. In this way Path identity type appeared in the core of type checker along with De Morgan algebra on built-in interval type.

```

inductive homotopy (lang : U) : U :=
| PathP: lang → lang → lang
| plam: name → lang → lang
| papp: lang → lang
| I
| zero
| one
| meet: lang → lang
| join: lang → lang
| neg: lang
| system: lang
| Partial: lang
| transp: lang → lang
| hcomp: lang
| Sub: lang
| inc: lang
| ouc: lang

```

► **Definition 4** (Cubical Presheaf \mathbb{I}). . *The identity types modeled with another presheaf, the presheaf on Lawvere category of distributive lattices (theory of De Morgan algebras) denoted with $\square \multimap \mathbf{I} : \square^{op} \rightarrow \text{Set}$.*

► **Definition 5** (Properties of \mathbf{I} . The presheaf \mathbf{I}). . *i) has to distinct global elements 0 and 1 (B_1); ii) $\mathbf{I}(I)$ has a decidable equality for each I (B_2); iii) \mathbf{I} is tiny so the path functor $X \mapsto X^{\mathbf{I}}$ has right adjoint (B_3).; iv) \mathbf{I} has meet and join (connections).*

Interval Pretypes. While having pretypes universe V with interval and associated De Morgan algebra $(\wedge, \vee, -, 0, 1, \mathbf{I})$ is enough to perform DNF normalization and proving some basic statements about path, including: contractability of singletons, homotopy transport, congruence, functional extensionality; it is not enough for proving β rule for Path type or path composition.

Generalized Transport. Generalized transport `transp` addresses first problem of deriving the computational β rule for Path types:

```

theorem Path $\beta$  (A : U) (a : A) (C : D A) (d: C a a (refl A a))
: Equ (C a a (refl A a)) d (J A a C d a (refl A a))
:=  $\lambda$  (A : U),  $\lambda$  (a : A),
 $\lambda$  (C :  $\Pi$  (x : A),  $\Pi$  (y : A), PathP (<_> A) x y  $\rightarrow$  U),
 $\lambda$  (d : C a a (<_> a)), <j> transp (<_> C a a (<_> a)) -j d

```

Transport is defined on fibrant types (only) and type checker should cover all the cases Note that `transpi (Pathj A v w) φ u0` case is relying on `comp` operation which depends on `hcomp` primitive. Here is given the first part of Simon Huber equations [5] for **transp**:

```

transpi N  $\varphi$  u0 = u0
transpi U  $\varphi$  A = A
transpi ( $\Pi$  (x : A), B)  $\varphi$  u0 v = transpi B(x/w)  $\varphi$  (u0 w(i/0))
transpi ( $\Sigma$  (x : A), B)  $\varphi$  u0 = (transpi A  $\varphi$  (u0.1), transpi B(x/v)  $\varphi$  (u0.2))
transpi (Pathj v w)  $\varphi$  u0 = <j> compi A [ $\phi$  u0 j, (j=0)  $\mapsto$  v, (j=1)  $\mapsto$  w] (u0 j)
transpi (Glue [ $\varphi \mapsto$  (T,w)] A)  $\psi$  u0 = glue [ $\phi$ (i/1)  $\mapsto$  t'₁] a'₁ : B(i/1)

```

Partial Elements. In order to explicitly define `hcomp` we need to specify `n`-cubes where some faces are missing. Partial primitives `isOne`, `1=1` and `UIP` on pretypes are derivable in Anders due to landing strict equality `Id` in `V` universe. The idea is that `(Partial A r)` is the type of cubes in `A` that are only defined when `IsOne r` holds. `(Partial A r)` is a special version of the function space `IsOne r → A` with a more extensional equality: two of its elements are considered judgmentally equal if they represent the same subcube of `A`. They are equal whenever they reduce to equal terms for all the possible assignment of variables that make `r` equal to `1`.

```
def Partial' (A : U) (i : I) := Partial A i
def isOne : I → V := Id I 1
def 1=>1 : isOne 1 := ref 1
def UIP (A : V) (a b : A) (p q : Id A a b) : Id (Id A a b) p q := ref p
```

Cubical Subtypes. For `(A : U) (i : I) (Partial A i)` we can define subtype `A [i ↦ u]`. A term of this type is a term of type `A` that is definitionally equal to `u` when `(IsOne i)` is satisfied. We have forth and back fusion rules `ouc (inc v) = v` and `inc (ouc v) = v`. Moreover, `ouc v` will reduce to `u 1=1` when `i=1`.

```
def sub' (A : U) (i : I) (u : Partial A i) : V := A [i ↦ u]
def inc' (A : U) (i : I) (a : A) : A [i ↦ [(i = 1) → a]] := inc A i a
def ouc' (A : U) (i : I) (u : Partial A i) (a : A [i ↦ u]) : A := ouc a
```

Homogeneous Composition. `hcomp` is the answer to second problem: with `hcomp` and `transp` one can express path composition, groupoid, category of groupoids (groupoid interpretation and internalization in type theory). One of the main roles of homogeneous composition is to be a carrier in [higher] inductive type constructors for calculating of homotopy colimits and direct encoding of CW-complexes. Here is given the second part of Simon Huber equations [5] for **hcomp**:

```
hcompi N [ϕ ↦ 0] 0 = 0
hcompi N [ϕ ↦ S u] (S u0) = S (hcompi N [ϕ ↦ u] u0)
hcompi U [ϕ ↦ E] A = Glue [ϕ ↦ (E(i/1), equivi E(i/1-i))] A
hcompi (Π (x : A), B) [ϕ ↦ u] u0 v = hcompi B(x/v) [ϕ ↦ u v] (u0 v)
hcompi (Σ (x : A), B) [ϕ ↦ u] u0 = (v(i/1), compi B(x/v) [ϕ ↦ u.2] u0.2)

hcompi (Pathj A v w) [ϕ ↦ u] u0
  = <j> hcompi A [ϕ ↦ u j, (j=0) ↦ v, (j=1) ↦ w] (u0 j)

hcompi (Glue [ϕ ↦ (T,w)] A) [ψ ↦ u] u0
  = glue [ϕ ↦ t1] a1
  = glue [ϕ ↦ u(i/1)] (unglue u(i/1))
  = u(i/1) : Glue [ϕ ↦ (T,w)] A
```

3.4 Strict Equality

To avoid conflicts with path equalities which live in fibrant universes strict equalities live in pretypes universes.

```
inductive strict (lang : U) : U :=
  | Id: name → lang
  | ref: lang → lang
  | idJ: lang → lang → lang
```

We use strict equality in `HTS` for pretypes and partial elements which live in `V`. The presheaf configuration with `Pi`, `Sigma` and `Id` is called **MLTT-73** [7]. The presheaf configuration with `Pi`, `Sigma`, `Id` and `Path` is called **HTS**.

3.5 Glue Types

The main purpose of Glue types is to construct a cube where some faces have been replaced by equivalent types. This is analogous to how `hcomp` lets us replace some faces of a cube by composing it with other cubes, but for Glue types you can compose with equivalences instead of paths. This implies the univalence principle and it is what lets us transport along paths built out of equivalences.

```
inductive glue (lang : U) : U :=
  | glue: lang → lang
  | unglue: lang → lang
```

Basic Fibrational HoTT core by Pelayo, Warren, and Voevodsky (2012).

```
def fiber (A B : U) (f : A → B) (y : B) : U := Σ (x : A), Path B y (f x)
def isEquiv (A B : U) (f : A → B) : U := Π (y : B), isContr (fiber A B f y)
def equiv (A B : U) : U := Σ (f : A → B), isEquiv A B f
def contrSingl (A : U) (a b : A) (p : Path A a b)
  : Path (Σ (x : A), Path A a x) (a, <i>a) (b, p) := <i> (p @ i, <j> p @ i ∨ j)
def idIsEquiv (A : U) : isEquiv A A (id A) :=
  λ (a : A), ((a, <i>a), λ (z : fiber A A (id A) a), contrSingl A a z.1 z.2)
def idEquiv (A : U) : equiv A A := (id A, isContrSingl A)
```

The notion of Univalence was discovered by Vladimir Voevodsky as forth and back transport between fibrational equivalence as contractability of fibers and homotopical multi-dimensional heterogeneous path equality. The `Equiv → Path` type is called Univalence type, where univalence intro is obtained by Glue type and `elim (Path → Equiv)` is obtained by sigma transport from constant map.

```
def univ-formation (A B : U) := equiv A B → PathP (<i> U) A B
def univ-intro (A B : U) : univ-formation A B := λ (e : equiv A B),
  <i> Glue B (∂ i) [(i = 0) → (A, e), (i = 1) → (B, idEquiv B)]
def univ-elim (A B : U) (p : PathP (<i> U) A B)
  : equiv A B := trans (<i> equiv A (p @ i)) 0 (idEquiv A)
def univ-computation (A B : U) (p : PathP (<i> U) A B)
  : PathP (<i> PathP (<i> U) A B) (univ-intro A B (univ-elim A B p)) p
  := <j i> Glue B (j ∨ ∂ i)
    [(i = 0) → (A, univ-elim A B p), (i = 1) → (B, idEquiv B),
     (j = 1) → (p @ i, univ-elim (p @ i) B (<k> p @ (i ∨ k)))]
```

Similar to Fibrational Equivalence the notion of Retract/Section based Isomorphism could be introduced as forth-back transport between isomorphism and path equality. This notion is somehow canonical to all cubical systems and is called Unimorphism here.

```
def iso-Form (A B : U) : U1 := iso A B → PathP (<i> U) A B
def iso-Intro (A B : U) : iso-Form A B :=
  λ (x : iso A B), isoPath A B x.f x.g x.s x.t
def iso-Elim (A B : U) : PathP (<i> U) A B → iso A B
  := λ (p : PathP (<i> U) A B),
    (coerce A B p, coerce B A (<i> p @ -i),
     trans-1-trans A B p, λ (a : A), <k> trans-trans-1 A B p a @-k, ★)
```

Orton-Pitts basis for univalence computability (2017):

```
def ua (A B : U) (p : equiv A B) : PathP (<i> U) A B := univ-intro A B p
def ua-β (A B : U) (e : equiv A B) : Path (A → B) (trans A B (ua A B e)) e.1
  := <i> λ (x : A), (idfun=idfun" B @ -i)
    ((idfun=idfun" B @ -i) ((idfun=idfun' B @ -i) (e.1 x)) )
```


3.6 de Rham Stack

Stack de Rham or Infinitesimal Shape Modality is a basic primitive for proving theorems from synthetic differential geometry. This type-theoretical framework was developed for the first time by Felix Cherubini under the guidance of Urs Schreiber. The Anders prover implements the computational semantics of the de Rham stack.

```

def  $\iota$  (A : U) (a : A) :  $\mathfrak{S}$  A :=  $\mathfrak{S}$ -unit a
def  $\mu$  (A : U) (a :  $\mathfrak{S}$  ( $\mathfrak{S}$  A)) :=  $\mathfrak{S}$ -join a

def is-coreduced (A : U) : U := isEquiv A ( $\mathfrak{S}$  A) ( $\iota$  A)
def  $\mathfrak{S}$ -coreduced (A : U) : is-coreduced ( $\mathfrak{S}$  A) := isoToEquiv
  ( $\mathfrak{S}$  A) ( $\mathfrak{S}$  ( $\mathfrak{S}$  A)) ( $\iota$  ( $\mathfrak{S}$  A)) ( $\mu$  A) ( $\lambda$  (x :  $\mathfrak{S}$  ( $\mathfrak{S}$  A)), <i>x) ( $\lambda$  (y :  $\mathfrak{S}$  A), <i>y)
def ind- $\mathfrak{S}\beta$  (A : U) (B :  $\mathfrak{S}$  A  $\rightarrow$  U) (f :  $\Pi$  (a : A),  $\mathfrak{S}$  (B ( $\iota$  A a))) (a : A)
  : Path ( $\mathfrak{S}$  (B ( $\iota$  A a))) (ind- $\mathfrak{S}$  A B f ( $\iota$  A a)) (f a) := <i> f a
def ind- $\mathfrak{S}$ -const (A B : U) (b :  $\mathfrak{S}$  B) (x :  $\mathfrak{S}$  A)
  : Path ( $\mathfrak{S}$  B) (ind- $\mathfrak{S}$  A ( $\lambda$  (i :  $\mathfrak{S}$  A), B) ( $\lambda$  (i : A), b) x) b := <i> b

```

Coreduced induction and its β -quation.

```

def  $\mathfrak{S}$ -ind (A : U) (B :  $\mathfrak{S}$  A  $\rightarrow$  U) (c :  $\Pi$  (a :  $\mathfrak{S}$  A),
  is-coreduced (B a)) (f :  $\Pi$  (a : A), B ( $\iota$  A a)) (a :  $\mathfrak{S}$  A) : B a
  := (c a (ind- $\mathfrak{S}$  A B ( $\lambda$  (x : A),  $\iota$  (B ( $\iota$  A x)) (f x)) a)).1.1
def  $\mathfrak{S}$ -ind $\beta$  (A : U) (B :  $\mathfrak{S}$  A  $\rightarrow$  U) (c :  $\Pi$  (a :  $\mathfrak{S}$  A),
  is-coreduced (B a)) (f :  $\Pi$  (a : A), B ( $\iota$  A a)) (a : A)
  : Path (B ( $\iota$  A a)) (f a) (( $\mathfrak{S}$ -ind A B c f) ( $\iota$  A a))
  := <i> sec-equiv (B ( $\iota$  A a)) ( $\mathfrak{S}$  (B ( $\iota$  A a))) ( $\iota$  (B ( $\iota$  A a)), c ( $\iota$  A a)) (f a) @-i

```

Geometric Modal HoTT Framework: Infinitesimal Proximity, Formal Disk, Formal Disk Bundle, Differential.

```

def  $\sim$  (X : U) (a x' : X) : U := Path ( $\mathfrak{S}$  X) ( $\iota$  X a) ( $\iota$  X x')
def  $\mathbb{D}$  (X : U) (a : X) : U :=  $\Sigma$  (x' : X),  $\sim$  X a x'
def inf-prox-ap (X Y : U) (f : X  $\rightarrow$  Y) (x x' : X) (p :  $\sim$  X x x')
  :  $\sim$  Y (f x) (f x') := <i>  $\mathfrak{S}$ -app X Y f (p @ i)
def d (X Y : U) (f : X  $\rightarrow$  Y) (x : X) ( $\varepsilon$  :  $\mathbb{D}$  X x)
  :  $\mathbb{D}$  Y (f x) := (f  $\varepsilon$ .1, inf-prox-ap X Y f x  $\varepsilon$ .1  $\varepsilon$ .2)
def  $T^\infty$  (A : U) : U :=  $\Sigma$  (a : A),  $\mathbb{D}$  A a
def inf-prox-ap (X Y : U) (f : X  $\rightarrow$  Y) (x x' : X) (p :  $\sim$  X x x')
  :  $\sim$  Y (f x) (f x') := <i>  $\mathfrak{S}$ -app X Y f (p @ i)
def d (X Y : U) (f : X  $\rightarrow$  Y) (x : X) ( $\varepsilon$  :  $\mathbb{D}$  X x)
  :  $\mathbb{D}$  Y (f x) := (f  $\varepsilon$ .1, inf-prox-ap X Y f x  $\varepsilon$ .1  $\varepsilon$ .2)
def  $T^\infty$ -map (X Y : U) (f : X  $\rightarrow$  Y) ( $\tau$  :  $T^\infty$  X) :  $T^\infty$  Y := (f  $\tau$ .1, d X Y f  $\tau$ .1  $\tau$ .2)

```

3.7 Higher Inductive Types

Anders currently don't support Lean-compatible generic inductive schemes definition. So instead of generic inductive schemes Anders supports well-founded trees (W-types). Basic data types like List, Nat, Fin, Vec are implemented as W-types in base library. As for higher inductive types Anders has Three-HIT basis (Coequalizer, HubSpoke and Colimit) to express other HIT.

References

- 1 Danil Annenkov, Paolo Capriotti, Nicolai Kraus, and Christian Sattler. Two-level type theory and applications, 2019. URL: <https://arxiv.org/pdf/1705.03307.pdf>.
- 2 Felix Cherubini. Cartan geometry in modal homotopy type theory, 2019. URL: <https://arxiv.org/pdf/1806.05966.pdf>.
- 3 Cyril Cohen, Thierry Coquand, Simon Huber, and Anders Mörtberg. Cubical type theory: A constructive interpretation of the univalence axiom, 2018. URL: <https://staff.math.su.se/anders.mortberg/papers/cubicalagda.pdf>.
- 4 Thierry Coquand, Simon Huber, and Anders Mörtberg. On higher inductive types in cubical type theory, 2017. URL: <https://staff.math.su.se/anders.mortberg/papers/cubicalhits.pdf>.
- 5 Simon Huber. On higher inductive types in cubical type theory, 2017. URL: <http://www.cse.chalmers.se/~simonhu/misc/hcomp.pdf>.
- 6 Martin-Löf. An intuitionistic theory of types, 1972.
- 7 Martin-Löf. An intuitionistic theory of types: Predicative part, 1975.
- 8 Martin-Löf. Intuitionistic type theory, 1980. URL: <https://raw.githubusercontent.com/michaelt/martin-lof/master/pdfs/Bibliopolis-Book-retypeset-1984.pdf>.
- 9 Andrea Vezzosi, Anders Mörtberg, and Andreas Abel. Cubical agda: A dependently typed programming language with univalence and higher inductive types, 2019. URL: <https://staff.math.su.se/anders.mortberg/papers/cubicalagda.pdf>.
- 10 Vladimir Voevodsky. A simple type system with two identity types, 2013. URL: <https://www.math.ias.edu/vladimir/sites/math.ias.edu.vladimir/files/HTS.pdf>.